
A BAYESIAN APPROACH FOR EVALUATING EQUIVALENCE OVER MULTIPLE GROUPS, AND COMPARISON WITH FREQUENTIST TOST


ELECTRONIC SUPPLEMENT 1. Example analysis using SAS scripts.
Analyses can be performed in any statistical platform that allows Bayesian calculations. Here it is illustrated for SAS (version 9.4).

One option for the analyses in SAS is to use PROC MCMC to generate the complete posterior distribution. In the leading example there are 4 sites participating. Non-informative uniform priors are used for the site means, and an inverse-gamma with very low parameter values as non-informative prior for the variance. In case of an informative prior, the syntax would be igamma(q1/2, scale=q2/2). 

In the SAS code below, the data set with observations is referred to as InputData, the response variable is referred to as Y, and the sites are numbered 1 to 4 using the variable Site. The code for PROC MCMC is as follows:

proc mcmc data=InputData nmc=50000000 nbi=5000 thin=50 outpost=postout;
 parms mu1=0 mu2=0 mu3=0 mu4=0;
 parms Var=1;
 prior mu1 mu2 mu3 mu4 ~ general(0);
 prior Var ~ igamma(0.001, scale=0.001);
 if Site=1 then mu=mu1;
 if Site=2 then mu=mu2;
 if Site=3 then mu=mu3;
 if Site=4 then mu=mu4;
 model Y ~ normal(mu , var=Var);
run;

Using the output data set of the posterior distribution the differences between the site means can be calculated.

data postout; set postout;
 EquivMargin=2;
 * Calculate individual differences between the site means ;
 delta_12=mu1-mu2;
 delta_13=mu1-mu3;
 delta_14=mu1-mu4;
 delta_23=mu2-mu3;
 delta_24=mu2-mu4;
 delta_34=mu3-mu4;
 * equivalence criterion is that the absolute difference is less than 
    the equivalence margin ;
 equiv_12=1; if abs(delta_12) >= EquivMargin then equiv_12=0;
 equiv_13=1; if abs(delta_13) >= EquivMargin then equiv_13=0;
 equiv_14=1; if abs(delta_14) >= EquivMargin then equiv_14=0;
 equiv_23=1; if abs(delta_23) >= EquivMargin then equiv_23=0;
 equiv_24=1; if abs(delta_24) >= EquivMargin then equiv_24=0;
 equiv_34=1; if abs(delta_34) >= EquivMargin then equiv_34=0;
 * only if all equiv-values are 1, then overall equivalence is obtained ;
 equiv_all=equiv_12*equiv_13*equiv_14*equiv_23*equiv_24*equiv_34;
run;

* summary statistics for the means and differences ;
proc means data=postout n mean std; by EquivMargin;
 var mu1 mu2 mu3 mu4 
     delta_12 delta_13 delta_14 delta_23 delta_24 delta_34;
run;

* posterior probabilities ;
proc means data=postout n mean; by EquivMargin;
 var equiv_12 equiv_13 equiv_14 equiv_23 equiv_24 equiv_34 equiv_all;
run;

The posterior distribution of the site means can also be generated by direct sampling from a multivariate t-distribution (equation 12). This can be done using the RandMVT-function in PROC IML. In this code the parameters q1 and q2 of the prior for the variance are assumed to be 0 (non-informative prior). In the syntax, the observed means as summarized in Table 2 of the main document are used, together with the value of  which equals 4.82+3.31+11.63+11.46 = 31.22.

proc iml;

 Mean = {32.26  33.62  32.18  33.68}; * observed site means ;
 NrObs = {8  9  18  20}; * numbers of observations per site ;

 Ntot=sum(NrObs); 
 Nsites=ncol(NrObs); 
 DF = Ntot - Nsites;

 PooledVar=31.22/DF; 
 * Using the sum of the corrected sums of squares per site ;

 M = j(Nsites,Nsites,0); * scale matrix of the multivariate t ;
  do i=1 to 4; M[i,i]=PooledVar/NrObs[i]; end; 

 print Mean, M, PooledVar, DF;

 call randseed(0);
 Nsim = 1000000; * simulation size ;

 x = RandMVT( Nsim, DF, Mean, M ); * drawing random numbers ;

 create SimData from x[colname={"mu1" "mu2" "mu3" "mu4"}];
 append from x; * make SAS data set ;

quit;

For further calculations, the data set SimData can be treated the same way as the postout data generated by PROC MCMC. 



